Determining Image Properties through ASP
by Mike Shaffer


In a recent small project, I needed a way to display a series of images in a square area (150x150 pixels). The images in question are uploaded by our customers, and we have no real control over what size or color depth (or even type) they are. While it's easy to force a resize of an image to, say, 150 pixels wide x 150 pixels high, you run the risk of stretching the image incorrectly in one dimension. And if you simply decide to force the width of the image to 150 and leave the height proportional (or vice-versa), then you may exceed the boundaries of 150 pixels. 

This is a problem. 

Fortunately, the solution is relatively easy. All we need is a tool that will report the dimensions (and color depth) of any image. I looked out on the web and found several that will do the trick. Of those, some were relatively inexpensive, and some were even free, but all of them were distributed as components. While there is certainly nothing wrong with that, I wanted something that could run entirely within an ASP page, without needing the ability to register a component on the server (since many people are unable to do so). 

Therefore, what follows is a set of routines (written in VBScript for your ASP pages) which will determine the height, width, and color depth of any GIF, JPG, BMP or PNG file. Additionally, these routines do not rely on the file extension to determine the graphic file-type. In other words, you can dynamically point this routine to any filename, regardless of name and extension, and determine if is a valid graphic file (actually, there is a very small chance [roughly one in 17 million] that a non-graphic file chosen at random will appear to be a graphics file, so it would be more appropriate to say that this routine determines PROBABLE graphic formats). 

Here are some possible uses for these routines: 

· intelligent image scaling (as mentioned above)

· Check user file upload, to see if the graphical image adheres to your requirements (e.g. make your upload function accept only files that are 640x480 or less and 256 colors)

· perform directory scans and gather information on your graphics files

· Show graphical directories (using thumbnails) on the web 

The comments in the routine should be sufficient to help you get started. The only real thing you need to understand is the format of the call to the main routine, gfxSpex: 

	xxx = gfxSpex(strFlnm, lngWidth, lngHeight, lngDepth, strImageType) 


where: 

· strFlnm => Filespec of file to read

· lngWidth => width of image

· lngHeight => height of image

· lngDepth => color depth (in number of colors)

· strImageType => type of image (e.g. GIF, BMP, etc.) 

The routine returns FALSE if the file was not recognized as a valid image type, and TRUE if it is 'probably' a valid image file. 

So there you have it, a complete new tool for your arsenal, with all VBScript source code, and best of all, IT'S FREE!

 Download the IMGSZ.ASP source code in text format
<%

  ':::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::

  ':::                                                             :::

  ':::  This routine will attempt to identify any filespec passed  :::

  ':::  as a graphic file (regardless of the extension). This will :::

  ':::  work with BMP, GIF, JPG and PNG files.                     :::

  ':::                                                             :::

  ':::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::

  ':::          Based on ideas presented by David Crowell          :::

  ':::                   (credit where due)                        :::

  ':::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::

  ':::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::

  ':::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::

  '::: blah blah blah blah blah blah blah blah blah blah blah blah :::

  '::: blah blah blah blah blah blah blah blah blah blah blah blah :::

  '::: blah blah     Copyright *c* MM,  Mike Shaffer     blah blah :::

  '::: blah blah      ALL RIGHTS RESERVED WORLDWIDE      blah blah :::

  '::: blah blah  Permission is granted to use this code blah blah :::

  '::: blah blah   in your projects, as long as this     blah blah :::

  '::: blah blah      copyright notice is included       blah blah :::

  '::: blah blah blah blah blah blah blah blah blah blah blah blah :::

  '::: blah blah blah blah blah blah blah blah blah blah blah blah :::

  ':::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::

  ':::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::

  ':::                                                             :::

  ':::  This function gets a specified number of bytes from any    :::

  ':::  file, starting at the offset (base 1)                      :::

  ':::                                                             :::

  ':::  Passed:                                                    :::

  ':::       flnm        => Filespec of file to read               :::

  ':::       offset      => Offset at which to start reading       :::

  ':::       bytes       => How many bytes to read                 :::

  ':::                                                             :::

  ':::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::

  option explicit

  function GetBytes(flnm, offset, bytes)

     Dim objFSO

     Dim objFTemp

     Dim objTextStream

     Dim lngSize

     Dim fsoForReading

     Dim strBuff

     on error resume next

     Set objFSO = CreateObject("Scripting.FileSystemObject")

     ' First, we get the filesize

     Set objFTemp = objFSO.GetFile(flnm)

     lngSize = objFTemp.Size

     set objFTemp = nothing

     fsoForReading = 1

     Set objTextStream = objFSO.OpenTextFile(flnm, fsoForReading)

     if offset > 0 then

        strBuff = objTextStream.Read(offset - 1)

     end if

     if bytes = -1 then

' Get All!

        GetBytes = objTextStream.Read(lngSize)  'ReadAll

     else

        GetBytes = objTextStream.Read(bytes)

     end if

     objTextStream.Close

     set objTextStream = nothing

     set objFSO = nothing

  end function

  ':::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::

  ':::                                                             :::

  ':::  Functions to convert two bytes to a numeric value (long)   :::

  ':::  (both little-endian and big-endian)                        :::

  ':::                                                             :::

  ':::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::

  function lngConvert(strTemp)

     lngConvert = clng(asc(left(strTemp, 1)) + ((asc(right(strTemp, 1)) * 256)))

  end function

  function lngConvert2(strTemp)

     lngConvert2 = clng(asc(right(strTemp, 1)) + ((asc(left(strTemp, 1)) * 256)))

  end function

  ':::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::

  ':::                                                             :::

  ':::  This function does most of the real work. It will attempt  :::

  ':::  to read any file, regardless of the extension, and will    :::

  ':::  identify if it is a graphical image.                       :::

  ':::                                                             :::

  ':::  Passed:                                                    :::

  ':::       flnm        => Filespec of file to read               :::

  ':::       width       => width of image                         :::

  ':::       height      => height of image                        :::

  ':::       depth       => color depth (in number of colors)      :::

  ':::       strImageType=> type of image (e.g. GIF, BMP, etc.)    :::

  ':::                                                             :::

  ':::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::

  function gfxSpex(flnm, width, height, depth, strImageType)

     dim strPNG, strGIF, strBMP

     dim strType, strTarget, strBuff

     dim lngSize

     dim flgFound

     strType = ""

     strImageType = "(unknown)"

     gfxSpex = False

     strPNG = chr(137) & chr(80) & chr(78)

     strGIF = "GIF"

     strBMP = chr(66) & chr(77)

     strType = GetBytes(flnm, 0, 3)

     if strType = strGIF then



' is GIF

        strImageType = "GIF"

        Width = lngConvert(GetBytes(flnm, 7, 2))

        Height = lngConvert(GetBytes(flnm, 9, 2))

        Depth = 2 ^ ((asc(GetBytes(flnm, 11, 1)) and 7) + 1)

        gfxSpex = True

     elseif left(strType, 2) = strBMP then

' is BMP

        strImageType = "BMP"

        Width = lngConvert(GetBytes(flnm, 19, 2))

        Height = lngConvert(GetBytes(flnm, 23, 2))

        Depth = 2 ^ (asc(GetBytes(flnm, 29, 1)))

        gfxSpex = True

     elseif strType = strPNG then


' Is PNG

        strImageType = "PNG"

        Width = lngConvert2(GetBytes(flnm, 19, 2))

        Height = lngConvert2(GetBytes(flnm, 23, 2))

        Depth = getBytes(flnm, 25, 2)

        select case asc(right(Depth,1))

           case 0

              Depth = 2 ^ (asc(left(Depth, 1)))

              gfxSpex = True

           case 2

              Depth = 2 ^ (asc(left(Depth, 1)) * 3)

              gfxSpex = True

           case 3

              Depth = 2 ^ (asc(left(Depth, 1)))  '8

              gfxSpex = True

           case 4

              Depth = 2 ^ (asc(left(Depth, 1)) * 2)

              gfxSpex = True

           case 6

              Depth = 2 ^ (asc(left(Depth, 1)) * 4)

              gfxSpex = True

           case else

              Depth = -1

        end select

     else

        strBuff = GetBytes(flnm, 0, -1)

' Get all bytes from file

        lngSize = len(strBuff)

        flgFound = 0

        strTarget = chr(255) & chr(216) & chr(255)

        flgFound = instr(strBuff, strTarget)

        if flgFound = 0 then

           exit function

        end if

        strImageType = "JPG"

        lngPos = flgFound + 2

        ExitLoop = false

        do while ExitLoop = False and lngPos < lngSize

           do while asc(mid(strBuff, lngPos, 1)) = 255 and lngPos < lngSize

              lngPos = lngPos + 1

           loop

           if asc(mid(strBuff, lngPos, 1)) < 192 or asc(mid(strBuff, lngPos, 1)) > 195 then

              lngMarkerSize = lngConvert2(mid(strBuff, lngPos + 1, 2))

              lngPos = lngPos + lngMarkerSize  + 1

           else

              ExitLoop = True

           end if

       loop

       '

       if ExitLoop = False then

          Width = -1

          Height = -1

          Depth = -1

       else

          Height = lngConvert2(mid(strBuff, lngPos + 4, 2))

          Width = lngConvert2(mid(strBuff, lngPos + 6, 2))

          Depth = 2 ^ (asc(mid(strBuff, lngPos + 8, 1)) * 8)

          gfxSpex = True

       end if

     end if

  end function

  ':::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::

  ':::     Test Harness                                              :::

  ':::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::

  ' To test, we'll just try to show all files with a .GIF extension in the root of C:

  dim objFSO, objF, objFC

  dim f1

  dim w, h, c

  dim strType

  Set objFSO = CreateObject("Scripting.FileSystemObject")

  Set objF = objFSO.GetFolder("c:\")

  Set objFC = objF.Files

  response.write "<table border=""0"" cellpadding=""5"">"

  For Each f1 in objFC

    if instr(ucase(f1.Name), ".GIF") then

       response.write "<tr><td>" & f1.name & "</td><td>" & f1.DateCreated & "</td><td>" & f1.Size & "</td><td>"

       if gfxSpex(f1.Path, w, h, c, strType) = true then

          response.write w & " x " & h & " " & c & " colors"

       else

          response.write "&nbsp;"

       end if

       response.write "</td></tr>"

    end if

  Next

  response.write "</table>"

  set objFC = nothing

  set objF = nothing

  set objFSO = nothing

%>

